

estimates notes — Add notes to estimation results

[Description](#)[Quick start](#)[Syntax](#)[Remarks and examples](#)[Also see](#)

Description

`estimates notes: text` adds a note to the current (active) estimation results.

`estimates notes` and `estimates notes list` list the current notes.

`estimates notes drop in noterange` eliminates the specified notes.

Quick start

Add “My note” to current estimation results

```
estimates notes: My note
```

List all notes for current estimation results

```
estimates notes list
```

Same as above

```
estimates notes
```

Drop notes 1 to 3 from current estimation results

```
estimates notes drop in 1/3
```

Drop last note applied to current estimation results

```
estimates notes drop in l
```

Syntax

```
estimates notes: text
```

```
estimates notes
```

```
estimates notes list [ in noterange ]
```

```
estimates notes drop in noterange
```

where *noterange* is # or #/# and where # may be a number, the letter f (meaning first), or the letter l (meaning last).

Remarks and examples

After adding or removing notes, if estimates have been stored, do not forget to store them again. If estimates have been saved, do not forget to save them again.

Notes are most useful when you intend to save estimation results in a file; see [R] [estimates save](#). For instance, after fitting a model, you might type

```
. estimates note: I think these are final
. estimates save lock2
```

and later when going through your files, you could type

```
. estimates use lock2
. estimates notes
1. I think these are final
```

Up to 9,999 notes can be attached to estimation results. If estimation results are important, we recommend that you add a note identifying the `.dta` dataset you used. The best way to do that is to type

```
. estimates notes: file 'c(filename)'
```

because `'c(filename)'` will expand to include not just the name of the file but also its full path; see [P] [creturn](#).

If estimation results took a long time to estimate—say, they were produced by `cmmprobit` or `gllamm` (see [CM] [cmmprobit](#) and <http://www.gllamm.org>)—it is also a good idea to add a data signature. A data signature takes less time to compute than reestimation when you need proof that you really have the right dataset. The easy way to do that is to type

```
. datasignature
74:12(71728):3831085005:1395876116
. estimates notes: datasignature reports 'r(datasignature)'
```

Now when you ask to see the notes, you will see

```
. estimates notes
1. I think these are final
2. file C:\project\one\pat4.dta
3. datasignature reports 74:12(71728):3831085005:1395876116
```

See [D] [datasignature](#).

Notes need not be positive. You might set a note to be, “I need to check that age is defined correctly.”

► Example 1

```
. use https://www.stata-press.com/data/r18/auto
(1978 automobile data)
. regress mpg weight displ if foreign
(output omitted)
. estimates notes: file 'c(filename)'
```

```
. datasignature
74:12(71728):3831085005:1395876116
. estimates notes: datasignature report 'r(datasignature)'
```

```
. estimates save foreign
file foreign.ster saved
```

```
. estimates notes list in 1/2
  1. file https://www.stata-press.com/data/r18/auto.dta
  2. datasignature report 74:12(71728):3831085005:1395876116
. estimates notes drop in 2
(1 note dropped)
. estimates notes
  1. file https://www.stata-press.com/data/r18/auto.dta
```



Also see

[R] [estimates](#) — Save and manipulate estimation results

Stata, Stata Press, and Mata are registered trademarks of StataCorp LLC. Stata and Stata Press are registered trademarks with the World Intellectual Property Organization of the United Nations. Other brand and product names are registered trademarks or trademarks of their respective companies. Copyright © 1985–2023 StataCorp LLC, College Station, TX, USA. All rights reserved.

